Exponential Matrix Examples

```
> restart:
  with(LinearAlgebra):
```

Construction of the problem

The resulting Jordan form

```
> J := subs( lambda=2, mu=4,
        <<lambda,0,0,0,0>|
        <1,lambda,0,0,0>|
        <0,0,mu,0,0>|
        <0,0,1,mu,0>|
        <0,0,0,1,mu>>) ;
```

$$J := \begin{bmatrix} 2 & 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 4 & 1 & 0 \\ 0 & 0 & 0 & 4 & 1 \\ 0 & 0 & 0 & 0 & 4 \end{bmatrix}$$

A change of coordinate

```
> T := <<1,0,1,2,0>|
        <1,1,0,3,0>|
        <0,0,1,0,0>|
        <0,0,1,-1,0>|
        <0,0,0,1,1>> ;
```

$$T := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 1 & 0 & 1 & 1 & 0 \\ 2 & 3 & 0 & -1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> T^(-1) ;
```

$$\begin{bmatrix} 1 & -1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ -3 & 0 & 1 & 1 & -1 \\ 2 & 1 & 0 & -1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> Determinant(T) ;
```

$$-1$$

A squared matrix

```
> A := simplify(T.J.T^(-1)) ;
```

$$A := \begin{bmatrix} 2 & 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 4 & 4 & -1 & 2 \\ -4 & 0 & 0 & 4 & -1 \\ 0 & 0 & 0 & 0 & 4 \end{bmatrix}$$

Using Jordan form

Copute exponential using Jordan form

Step 1: Evalute characteristic polynomial

```
> chpoly := Determinant(A-lambda*IdentityMatrix(5)) ;
```

$$chpoly := -(2-\lambda)^2 (-4+\lambda)^3$$

Step2: find eigenvalues

```
> lambda1 := 2 ;
  lambda2 := 4 ;
```

$$\lambda 1 := 2$$

$$\lambda 2 := 4$$

Step 3: find eigenvectors and generalized eigenvectors

```
> B1 := A-lambda1*IdentityMatrix(5) ;
```

$$B1 := \begin{bmatrix} 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 4 & 2 & -1 & 2 \\ -4 & 0 & 0 & 2 & -1 \\ 0 & 0 & 0 & 0 & 2 \end{bmatrix}$$

```
> RES := solve(convert( B1.<z||(1..5)>, list) ,{z||(1..5)});
```

$$RES := \{z2=0, z1=z3, z4=2\,z3, z3=z3, z5=0\}$$

find first eigenvector

```
> v1 := subs( z3=1,subs( RES, <z||(1..5)>)) ;
```

$$v1 := \begin{bmatrix} 1 \\ 0 \\ 1 \\ 2 \\ 0 \end{bmatrix}$$

find generalized eigenvector (A vk = lambda vk + v(k-1))

```
> RES := solve(convert( B1.<z||(1..5)>-v1, list) ,{z||(1..5)});
```

$$RES := \{z3=z3, z5=0, z2=1, z1=1+z3, z4=3+2\,z3\}$$

```
> v2 := subs( z3=0,subs( RES, <z||(1..5)>)) ;
```

$$v2 := \begin{bmatrix} 1 \\ 1 \\ 0 \\ 3 \\ 0 \end{bmatrix}$$

No solution, end of chain

```
> RES := solve(convert( B1.<z||(1..5)>-v2, list) ,{z||(1..5)});
```

$$RES :=$$

find second eigenvector

```
> B2 := A-lambda2*IdentityMatrix(5) ;
```

$$B2 := \begin{bmatrix} -2 & 1 & 0 & 0 & 0 \\ 0 & -2 & 0 & 0 & 0 \\ 0 & 4 & 0 & -1 & 2 \\ -4 & 0 & 0 & 0 & -1 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> RES := solve(convert( B2.<z||(1..5)>, list) ,{z||(1..5)});
```

$$RES := \{z2 = 0, z4 = 0, z1 = 0, z3 = z3, z5 = 0\}$$

```
> v3 := subs( z3=1,subs( RES, <z||(1..5)>)) ;
```

$$v3 := \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}$$

find generalized eigenvector (A vk = lambda vk + v(k-1))

```
> RES := solve(convert( B2.<z||(1..5)>-v3, list) ,{z||(1..5)});
```

$$RES := \{z2 = 0, z1 = 0, z3 = z3, z4 = -1, z5 = 0\}$$

```
> v4 := subs( z3=0,subs( RES, <z||(1..5)>)) ;
```

$$v4 := \begin{bmatrix} 0 \\ 0 \\ 0 \\ -1 \\ 0 \end{bmatrix}$$

find generalized eigenvector (A vk = lambda vk + v(k-1))

```
> RES := solve(convert( B2.<z||(1..5)>-v4, list) ,{z||(1..5)});
```

$$RES := \{z2 = 0, z1 = 0, z3 = z3, z5 = 1, z4 = 2\}$$

```
> v5 := subs( z3=0,subs( RES, <z||(1..5)>)) ;
```

$$v5 := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 2 \\ 1 \end{bmatrix}$$

No solution, end of chain

```
> RES := solve(convert( B2.<z||(1..5)>-v5, list) ,{z||(1..5)});
```

$$RES :=$$

Step 4: compute matrix T for Jordan canonical form

```
> T := <v1|v2|v3|v4|v5> ;
```

$$T := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 1 & 0 & 1 & 0 & 0 \\ 2 & 3 & 0 & -1 & 2 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

Verification and Jordan matrix calculation

```
> J := T^(-1).A.T ;
```

$$J := \begin{bmatrix} 2 & 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 4 & 1 & 0 \\ 0 & 0 & 0 & 4 & 1 \\ 0 & 0 & 0 & 0 & 4 \end{bmatrix}$$

Step 5: Block estraction and exponential matrix of the single block

```
> J1 := J[1..2,1..2] ;
  J2 := J[3..5,3..5] ;
```

$$J1 := \begin{bmatrix} 2 & 1 \\ 0 & 2 \end{bmatrix}$$

$$J2 := \begin{bmatrix} 4 & 1 & 0 \\ 0 & 4 & 1 \\ 0 & 0 & 4 \end{bmatrix}$$

```
> ExpJ1 := exp(lambda1)*<<1,0>|<1,1>> ;
```

$$ExpJ1 := \begin{bmatrix} e^2 & e^2 \\ 0 & e^2 \end{bmatrix}$$

```
> ExpJ2 := exp(lambda2)*<<1,0,0>|<1,1,0>|<1/2,1,1>> ;
```

$$ExpJ2 := \begin{bmatrix} e^4 & e^4 & \frac{1}{2} e^4 \\ 0 & e^4 & e^4 \\ 0 & 0 & e^4 \end{bmatrix}$$

blockj composition

```
> ExpJ := Matrix(5,5) ;
```

$$ExpJ := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> ExpJ[1..2,1..2] := ExpJ1 ;
  ExpJ[3..5,3..5] := ExpJ2 ;
```

$$ExpJ_{1..2,\,1..2} := \begin{bmatrix} e^2 & e^2 \\ 0 & e^2 \end{bmatrix}$$

$$ExpJ_{3..5,\,3..5} := \begin{bmatrix} e^4 & e^4 & \frac{1}{2} e^4 \\ 0 & e^4 & e^4 \\ 0 & 0 & e^4 \end{bmatrix}$$

```
> ExpJ;
```

$$\begin{bmatrix} e^2 & e^2 & 0 & 0 & 0 \\ 0 & e^2 & 0 & 0 & 0 \\ 0 & 0 & e^4 & e^4 & \frac{1}{2} e^4 \\ 0 & 0 & 0 & e^4 & e^4 \\ 0 & 0 & 0 & 0 & e^4 \end{bmatrix}$$

Step 6: back transformation for exponential matrix computation

```
> ExpA := T.ExpJ.T^(-1) ;
```

$$ExpA := \begin{bmatrix} e^2 & e^2 & 0 & 0 & 0 \\ 0 & e^2 & 0 & 0 & 0 \\ e^2 + e^4 & 2\, e^4 & e^4 & -e^4 & \frac{5}{2} e^4 \\ 2\, e^2 - 2\, e^4 & 3\, e^2 - e^4 & 0 & e^4 & -e^4 \\ 0 & 0 & 0 & 0 & e^4 \end{bmatrix}$$

Compute exponential using Caley-Hamilton

Step 1: Evalute characteristic polynomial

```
> chpoly := Determinant(A-lambda*IdentityMatrix(5)) ;
```

$$chpoly := -(2-\lambda)^2(-4+\lambda)^3$$

Step2: find eigenvalues

```
> lambda1 := 2 ;
  lambda2 := 4 ;
```

$$\lambda 1 := 2$$

$$\lambda 2 := 4$$

Step 3: compute interpolation condition

```
> r := lambda -> r0+r1*lambda+r2*lambda^2+r3*lambda^3+r4*
  lambda^4 ;
```

$$r := \lambda \rightarrow r0 + r1\,\lambda + r2\,\lambda^2 + r3\,\lambda^3 + r4\,\lambda^4$$

```
> EQ1 := r(lambda1)          - exp(lambda1) ;
  EQ2 := D(r)(lambda1)       - exp(lambda1) ;
  EQ3 := r(lambda2)          - exp(lambda2) ;
  EQ4 := D(r)(lambda2)       - exp(lambda2) ;
  EQ5 := (D@@2)(r)(lambda2)  - exp(lambda2) ;
```

$$EQ1 := r0 + 2\,r1 + 4\,r2 + 8\,r3 + 16\,r4 - e^2$$

$$EQ2 := r1 + 4\,r2 + 12\,r3 + 32\,r4 - e^2$$

$$EQ3 := r0 + 4\,r1 + 16\,r2 + 64\,r3 + 256\,r4 - e^4$$

$$EQ4 := r1 + 8\,r2 + 48\,r3 + 256\,r4 - e^4$$

$$EQ5 := 2\,r2 + 24\,r3 + 192\,r4 - e^4$$

```
> RES := solve( {EQ||(1..5)}, {r||(0..4)} ) ;
```

$$RES := \left\{ r4 = \frac{1}{16}\,e^4 - \frac{5}{16}\,e^2,\ r3 = -\frac{3}{4}\,e^4 + \frac{17}{4}\,e^2,\ r0 = 5\,e^4 - 32\,e^2,\ r1 = -7\,e^4 + 44\,e^2,\ r2 = \frac{7}{2}\,e^4 - 21\,e^2 \right\}$$

```
> ExpCH := simplify(subs( RES, r(A) )-r(0))+r(0)*IdentityMatrix
  (5) ;
```

$$ExpCH := \begin{bmatrix} e^2 & e^2 & 0 & 0 & 0 \\ 0 & e^2 & 0 & 0 & 0 \\ e^2 + e^4 & 2\,e^4 & e^4 & -e^4 & \frac{5}{2}\,e^4 \\ 2\,e^2 - 2\,e^4 & 3\,e^2 - e^4 & 0 & e^4 & -e^4 \\ 0 & 0 & 0 & 0 & e^4 \end{bmatrix}$$

```
> evalf(ExpCH) ;
```

$$\begin{bmatrix} 7.389056099 & 7.389056099 & 0. & 0. & 0. \\ 0. & 7.389056099 & 0. & 0. & 0. \\ 61.98720613 & 109.1963001 & 54.59815003 & -54.59815003 & 136.4953751 \\ -94.41818790 & -32.43098173 & 0. & 54.59815003 & -54.59815003 \\ 0. & 0. & 0. & 0. & 54.59815003 \end{bmatrix}$$

>